Symbolic Integration

Victor Adamchik

Carnegie Mellon University

Hermite-Ostrogradsky's Algorithm

Given

$$\int \frac{p}{q} = \int \frac{p}{q_1 * q_2^2 * \ldots q_m^m}$$

where $\deg(p) < \deg(q)$ and $\text{GCD}(p, q) = 1$. Hermite-Ostrogradsky's algorithm reduces exponents of each q_k to 1

$$\int \frac{1}{q_1 * q_2^2 * \ldots q_m^{m}} \to \int \frac{r_1}{q_1 * q_2^2 * \ldots q_m^{m-1}} \to \ldots \to \int \frac{r_k}{q_1 * q_2 * \ldots q_m}$$

The algorithm proceeds as follows:

$$\int \frac{p}{q} = \frac{1}{1-m} \frac{a}{v^{m-1}} + \frac{1}{1-m} \int \frac{b\,(1-m) - u\,a'}{q_1 * q_2^2 * \ldots q_m^{m-1}}$$

where

$$q = q_1 * q_2^2 * \ldots q_m^m$$

$$u = q_1 * q_2^2 * \ldots q_{m-1}^{m-1}$$

$$p = u\,q_m'\,a + q_m\,b, \qquad \deg(a) \leq \deg(v) - 1$$

- **Example**

$$\int \frac{x^2 - \frac{2}{3}x - \frac{2}{3}}{(x^3 + x + 1)^2}\,dx$$

```
p = x^2 - 2x/3 - 2/3;
q = (x^3 + x + 1)^2;
v = x^3 + x + 1;
m = 2;
u = q/v^m;
```

```
{p1, p2} = PolynomialExtendedGCD[u*D[v, x], v][[2]]
```

$$\left\{\frac{4}{31} - \frac{9x}{31} + \frac{6x^2}{31}, \frac{27}{31} - \frac{18x}{31}\right\}$$

Reduce *p1* $*$ *p* and compute *a* and *b*

```
b = Expand[PolynomialQuotient[p1*p, v, x] *u*D[v, x] + p2*p]
```

-1

```
a = PolynomialRemainder[p1*p, v, x]
```

$$\frac{1}{3} + \frac{x}{3}$$

The integral

$$\int \frac{p}{q} = \frac{1}{1-m} * \frac{a}{v^{m-1}} + \frac{1}{1-m} \int \frac{b*(1-m) - u*a'}{u*v^{m-1}}$$

```
(b (1 - m) - u D[a, x])/(u v^(m-1)) // Simplify
```

$$\frac{2}{3\,(1 + x + x^3)}$$

```
    a
─────────── // Simplify
(1 - m) v^(m-1)
```

```
    1 + x
- ───────────────
  3 (1 + x + x^3)
```

or

$$\int \frac{x^2 - \frac{2}{3}x - \frac{2}{3}}{(x^3 + x + 1)^2}\,dx = -\frac{1+x}{3\,(1+x+x^3)} - \frac{2}{3}\int \frac{1}{1+x+x^3}\,dx$$

The integral in the right hand side is computed by a different algorithm.

Rothstein-Trager's Algorithm

Consider

$$\int \frac{dx}{x^3 + x} = \int \frac{dx}{x(x+i)\,(x-i)}$$

By decomposition

$$\frac{1}{x\,(x+i)\,(x-i)} = \frac{1}{x} - \frac{1}{2\,(x-i)} - \frac{1}{2\,(x+i)}$$

we get

$$\int \frac{1}{x^3+x}\,dx = \log(x) - \frac{1}{2}\log(x-i) - \frac{1}{2}\log(x+i)$$

However, the integral can be expressed without requiring the extention to $Q[i]$

```
Integrate[1/(x^3 + x), x]
```

```
Log[x] - 1/2 Log[1 + x^2]
```

Rothstein-Trager's Algorithm deals with the logarithmic part of the integral

$$\int \frac{p(x)}{q(x)}\,dx$$

where $\deg(p) < \deg(q)$ and q is monic and squarefree. The algorithm finds the smallest algebraic field extension of F.

Definition. A *splitting field* of P over $F[x]$ is the smallest algebraic field extension of F containing all roots of P.

We start with formal factoring the denominator q over its splitting field

$$\int \frac{p}{q}\,dx = \int \frac{p(x)}{\prod_{k=1}^{n}(x-\alpha_k)}\,dx = \sum_{k=1}^{n} \beta_k * \log(x-\alpha_k)$$

and figure out the way of finding coefficients β_k. Expanding p/q in the Laurent series at $x = \alpha_r$

$$\frac{p}{q} = \frac{p(x)}{\prod_{k=1}^{n}(x-\alpha_k)} = \frac{c}{x-\alpha_r} + c_0 + c_1(x-\alpha_r) + \ldots$$

How do you find c?

$$\frac{p(x)}{\prod_{k=1}^{n}(x-\alpha_k)} - \frac{c}{x-\alpha_r} = c_0 + c_1(x-\alpha_r) + \ldots$$

Multiply this by $x - \alpha_r$

$$\frac{p(x)}{\prod_{\substack{k=1 \\ k\neq r}}^{n}(x-\alpha_k)} - c = c_0\,(x-\alpha_r) + c_1(x-\alpha_r)^2 + \ldots$$

to obtain (after replacing x by α_r)

$$c = \frac{p(\alpha_r)}{\prod_{\substack{k=1 \\ k\neq r}}^{n}(\alpha_r-\alpha_k)}$$

This could be further simplified. Let us work out the derivative of the denominator

$$q'(x) = \left(\prod_{k=1}^{n}(x-\alpha_k)\right)'$$

$$q'(x) = (x-\alpha_r)' * \prod_{\substack{k=1 \\ k\neq r}}^{n}(x-\alpha_k) + (x-\alpha_r) * \left(\prod_{\substack{k=1 \\ k\neq r}}^{n}(x-\alpha_k)\right)'$$

Setting $x = \alpha_r$, yeilds

$$q'(\alpha_r) = \prod_{\substack{k=1 \\ k\neq r}}^{n}(x-\alpha_k)$$

Therefore,

$$c = \beta_r = \frac{p(\alpha_r)}{q'(\alpha_r)} \tag{1}$$

and the integral can be rewritten as

$$\int \frac{p(x)}{q(x)}\, dx = \sum_{k=1}^{n} \frac{p(\alpha_k)}{q'(\alpha_k)} * \log(x - \alpha_k) \tag{2}$$

The problem is reduced to finding all distinct roots of $q(x) = 0$.

Consider the equation (2)

$$p(\alpha_p) - \beta_p * q'(\alpha_p) = 0$$

Since we have to find a correspondent β_p for each $p = 1, 2, \ldots, n$, we have an equivalence

$$q(\alpha_r) = 0 \implies p(\alpha_r) - \beta_r\, q'(\alpha_r) = 0$$

Hence, both these equations have common roots

$$\begin{cases} q(x) = 0 \\ p(x) - z * q'(x) = 0 \end{cases}$$

To determine whether or not two polynomials have a non-trivial common factor we will be using a theory of resultants.

Theorem **(Rothstein-Trager's Algorithm)**. *Let $p, q \in F[x]$, $GCD(p, q) = 1$, $deg(p) < deg(q)$, q is monic and square-free. Then*

$$\int \frac{p}{q} = \sum_{k=1}^{n} \beta_k * log(d_k)$$

where β_k are the distinct roots of

$$R(z) = res_x(q(x),\ p(x) - z * q'(x))$$

and d_k are the polynomials

$$d_k = GCD(q(x),\ p(x) - c_k * q'(x))$$

■ Example

$$\int \frac{x + 2}{x(x^2 + x + 1)}\, dx$$

Compute the resultant

```
Resultant[x (1 + x + x^2), (2 + x) - z ∂x (x (1 + x + x^2)), x]
```

```
(2 - z) (3 + 6 z + 3 z^2)
```

Find destinct roots

```
z /. Solve[% == 0, z]
```

```
{-1, -1, 2}
```

No algebraic extension is required. $c_1 = 2, c_2 = -1$.

The log arguments are computed by using GCD

```
PolynomialGCD[x (1 + x + x^2), x + 2 - 2 ∂x (x (1 + x + x^2))]
```

```
x
```

```
PolynomialGCD[x (1 + x + x^2), x + 2 + ∂x (x (1 + x + x^2))]
```

```
1 + x + x^2
```

Thus,

$$\int \frac{x+2}{x\,(x^2+x+1)}\,dx = 2\log(x) - \log\left(1+x+x^2\right)$$

■ **Example**

$$\int \frac{1}{x^2-3}\,dx$$

```
Resultant[x^2 - 3, 1 - z ∂x (x^2 - 3), x]
```

```
1 - 12 z^2
```

An algebraic extension is required

```
z /. Solve[1 - 12 z^2 == 0, z]
```

$$\left\{-\frac{1}{2\sqrt{3}}, \frac{1}{2\sqrt{3}}\right\}$$

```
PolynomialGCD[1 - ∂x(x^2 - 3)/(2 √3), x^2 - 3, Extension → √3]
```

$\sqrt{3} - x$

```
PolynomialGCD[1 + ∂x(x^2 - 3)/(2 √3), x^2 - 3, Extension → √3]
```

$\sqrt{3} + x$

The integral

$$\int \frac{dx}{x^2-3} = \frac{1}{2\sqrt{3}} * \log\left(\sqrt{3} - x\right) - \frac{1}{2\sqrt{3}} * \log\left(\sqrt{3} + x\right)$$

■ The Czichowski algorithm

Czichowski's idea is to use the Gröbner basis. Let us consider

$$\int \frac{1}{(x-2)(x^2-1)} dx$$

Compute the Groebner basis

```
{p1, p2} = GroebnerBasis[{(x - 2) (x^2 - 1), 1 - z D[(x - 2) (x^2 - 1), x]}, {x, z}]
```

Find roots

```
roots = z /. Solve[p1 == 0, z]
```

```
Inner[Power, {1, 2, 3}, {a, b, c}, Plus]
```

```
Inner[#1 Log[#2] &, roots, p2 /. z -> roots, Plus]
```

The integral

$$\int \frac{1}{(x-2)(x^2-1)}\,dx = \frac{1}{6}\log(5\,x+5) - \frac{1}{2}\log(5\,x-5\,) + \frac{1}{3}\log(5\,x-10)$$

References

E. Hermite, Sur l'intégration des fractions rationelles, *Nouvelles Annales de Mathématiques*, **11**(1872), 145-148.

M. W. Ostrogradsky, De l'intégration des fractions rationelles, *Bulletin de la Classe Physico-Mathématiques de l'Académie Impériale des Sciences de St. Pétersburg*, IV, 1845, pp145-167, 286-300.

M. Bronstein, *Symbolic Integration - Transcendental Functions,* Algorithms and Computations in Mathematics, Vol 1, Springer-Verlag, 1996.